```
> restart;  Digits:=14;
```

$$\text{Digits} := 14$$

```
> unprotect(gamma); unprotect(lambda);
> assume(0<k); # assume( j in {0,1} );
> kappa:= lambda; # mean reverting rate
  theta:= v1; # long term variance
  epsilon:= eta; # vol of vol
  rho:= rho; # correlation
  V:= v2; # current variance
  tau:= tau; # remaining time
```

$$\kappa := \lambda$$

$$\theta := v1$$

$$\varepsilon := \eta$$

$$\rho := \rho$$

$$V := v2$$

$$\tau := \tau$$

```
> tau:=5.0;v1:=0.07;v2:=0.07;lambda:=1.0;eta:=0.5;rho:=-0.5;
```

$$\tau := 5.0$$

$$v1 := 0.07$$

$$v2 := 0.07$$

$$\lambda := 1.0$$

$$\eta := 0.5$$

$$\rho := -0.5$$

```
> rates:=0.15;
  fwd:=100.0*exp(rates*tau); strike:=100;
  X:='ln(fwd/strike)';  X:=evalf(%): X:=0.75;
```

$$\text{rates} := 0.15$$

$$\text{fwd} := 211.70000166127$$

$$\text{strike} := 100$$

$$X := \ln\left(\frac{\text{fwd}}{\text{strike}}\right)$$

$$0.75000000000001$$

$$X := 0.75$$

```
> u:= 'kappa-rho*epsilon/2';
  zeta:= 'sqrt( k^2*epsilon^2*(1-rho^2) + 2*I*k*rho*epsilon*u + u^2 +
  epsilon^2/4)';
    zeta:=evalf(zeta):
  psiPlus:= '-(u+I*k*rho*epsilon) + zeta';
  psiMinus:= '+(u+I*k*rho*epsilon) + zeta';
  ``;
```

$$u := \kappa - \frac{\rho\,\varepsilon}{2}$$

$$\zeta := \sqrt{k^2\,\varepsilon^2\,(1-\rho^2) + 2\,I\,k\,\rho\,\varepsilon\,u + u^2 + \frac{\varepsilon^2}{4}}$$

$$\text{psiPlus} := -u - k\,\rho\,\varepsilon\,I + \zeta$$

$$\text{psiMinus} := u + k\,\rho\,\varepsilon\,I + \zeta$$

```
> A:= '-kappa*theta/epsilon^2*( psiPlus*tau+2*ln(
  (psiMinus+psiPlus*exp(-zeta*tau))/2/zeta ) )'; ``;
  B:= '-(k^2+1/4)*(1-exp(-zeta*tau))/(psiMinus+psiPlus*exp(-zeta*tau))'; ``;
```

$$A := -\frac{\kappa\,\theta\left(\text{psiPlus}\,\tau + 2\ln\left(\frac{1}{2}\,\frac{\text{psiMinus} + \text{psiPlus}\,\mathbf{e}^{(-\zeta\tau)}}{\zeta}\right)\right)}{\varepsilon^2}$$

$$B := -\frac{\left(k^2 + \frac{1}{4}\right)\left(1 - \mathbf{e}^{(-\zeta\tau)}\right)}{\text{psiMinus} + \text{psiPlus}\,\mathbf{e}^{(-\zeta\tau)}}$$

```
> Q:= 'exp( (-I*k+1/2)*X + A + B*V )'; # indets(%,atomic): indets(%);
  Q:= evalf(%):
```

$$Q := \mathbf{e}^{((-k\,I + 1/2)\,X + A + B\,V)}$$

```
> evalf( Q / (k^2+1/4) ): # simplify(%) assuming 0<k:
  evalc(%):
  Re(%):
  theRealIntegrand:=%:
> cutOff:=128;
```

$$\text{cutOff} := 128$$

```
> st:=time():
  'Int(theRealIntegrand,k=0..cutOff)';:; %: P:= Re(evalf(%));
  `seconds`=time()-st;
  strike/Pi*P: evalf(%):
  fwd-%;
  `discounted`=% * exp(-rates*tau);
```

$$\int_0^{\text{cutOff}} \text{theRealIntegrand}\,dk$$

$$P := 2.9623041248051$$

$$\text{seconds} = 1.576$$

$$117.40693278042$$

$$\text{discounted} = 55.459108105382$$

```
> st:=time():
  'Int(Re( Q / (k^2+1/4) ),k=0..infinity)'; %: P:= Re(evalf(%));
  `seconds`=time()-st;
  strike/Pi*P: evalf(%):
  fwd-%;
  `discounted`=% * exp(-rates*tau);
```

$$\int_0^{\infty} \Re\left(\frac{Q}{k^2 + \frac{1}{4}}\right) dk$$

$$P := 2.9623041248051$$

$$\text{seconds} = 1.304$$

$$117.40693278042$$

$$\text{discounted} = 55.459108105382$$